

## Utilitaire pour Super-Oric "RCLORES1.COM" en Basic 1.1 Redéfinition des caractères en LORES 1

par André

La dernière fois, nous avons vu comment automatiser le calcul des DATA à incorporer dans la commande DEFCHAR pour redéfinir les caractères en LORES 0. Aujourd'hui, voici le programme équivalent pour LORES 1. Ce programme devrait grandement vous faciliter la vie lors de la programmation pour Super-Oric. Je rappelle qu'il n'est pas nécessaire de posséder une console Snes pour développer et jouer sur Super-Oric. Il est tout à fait possible de développer sous Euphoric (en configuration Atmos), voire même simplement avec l'utilitaire Txt2basic, de confectionner automatiquement les fichiers-cartouches grâce à l'utilitaire Builcart.exe (voir mag 198 pages 4 & 11) et de tester avec l'un des émulateurs de Snes (adresses:

voir le mag 173 page 6). Toutefois, si vous disposez d'un clavier pour Super-Oric et de tout le matériel approprié, vous pouvez aussi développer sur Super-Oric réel. En effet, **cet utilitaire, qui est écrit en basic 1.1, tourne aussi bien sur Atmos que sur Super-Oric.** Dans tous les cas, vous devrez concevoir vos caractères redéfinis en coloriant les cases d'une grille 8x8 à l'aide de crayons de couleurs (16 au maximum), puis d'attribuer un n° à chaque couleur (de 0 à 15 maximum) et de taper la liste de ces n° dans RCLORES1. Le programme effectue des calculs à la fin de chaque ligne de 8 valeurs et à la fin affiche la ligne DEFCHAR à incorporer dans votre programme pour Super-Oric. Bonne programmation...

```

1000 ' *** REDEFCHAR ***
1010 ' (c) Andr{
1020 ' V1811061140
1030 ' SAVEU"RCLORES1"
1040 DIM L(7,7):'Tableau des couleurs des 8 pixels des 8 lignes
1050 DIM C(7,7,3):'Plans 0 @ 3 des couleurs des 8x8 pixels du
Caract}re
1060 DIM W(16):' Liste des Words pour la ligne DEFCHAR
1070 ' *** Titre
1080 CLS:PRINT"REDEFCHAR":PRINT"pour Super-Oric"
1090 PRINT"(C) Andr{ Ch{ramy 2006":PRINT
1100 PRINT"Code ASCII du caract}re"
1110 PRINT"(de 0 @ 1023)";:INPUT CA
1120 PRINT"Pour chacune des 8 lignes,"
1130 PRINT"entrez la couleur des 8 pixels":PRINT"de red{finition"
1140 PRINT"Valeurs de 0 @15 en LORES1"
1150 ' *** D{but de boucle pour 8 lignes
1160 FOR LIGNE=0 TO 7
1170 ' *** Remplissage des couleurs des pixels de la ligne
1180 PRINT:PRINT"Couleurs pour la ligne N";LIGNE
1190 FOR P=0 TO 7:' 8 Pixels
1200 PRINT"Pixel"P" ";:INPUT L(P,LIGNE)
1210 NEXT P
1220 ' *** Chaque bit est plac{ dans un plan distinct
1230 ' *** et se voit attribu{ une valeur V d{pendant de sa posi-
tion
1240 ' *** dans la colonne de d{finition du caract}re
1250 V=8:' Initialisation pour colonne 0
1260 M=1:' Masque pour b0
1270 FOR P=0 TO 7:' Pixels de 0 @ 7
1280 CP=L(P,LIGNE):'Lecture Couleur Pixel

```

```

REDEFCHAR                                CAPS
pour Super-Oric
(C) André Chéramy 2006

Code ASCII du caractère
(de 0 à 1023) ? 66
Pour chacune des 8 lignes,
entrez la couleur des 8 pixels
de redéfinition
Valeurs de 0 à 15 en LORES1

Couleurs pour la ligne N 0
Pixel 0 ? 0
Pixel 1 ? 1
Pixel 2 ? 2
Pixel 3 ? 3
Pixel 4 ? 4
Pixel 5 ? 5
Pixel 6 ? 6
Pixel 7 ? 7

Couleurs pour la ligne N 1
Pixel 0 ? 0
Pixel 1 ? 1
Pixel 2 ? 10
Pixel 3 ? 7

```

```

Pixel 7 ? 15                                CAPS

Couleurs pour la ligne N 6
Pixel 0 ? 0
Pixel 1 ? 1
Pixel 2 ? 2
Pixel 3 ? 3
Pixel 4 ? 4
Pixel 5 ? 5
Pixel 6 ? 6
Pixel 7 ? 7

Couleurs pour la ligne N 7
Pixel 0 ? 0
Pixel 1 ? 9
Pixel 2 ? 10
Pixel 3 ? 11
Pixel 4 ? 12
Pixel 5 ? 13
Pixel 6 ? 14
Pixel 7 ? 15

DEFCHAR#42,#3355,#3355,#3355,#3355,#33
55,#3355,#3355,#3355,#F,#FF0F,#F,#FF0F
#F,#FF0F,#F,#FF0F
Ready

```

```

1290 FOR B=0 TO 3:' bits de 0 @ 3
1300 IF(CP AND M) > 0 THEN C(0,P,B)=V ELSE C(0,P,B)=0
1310 M=M*2
1320 NEXT B
1330 M=1:' Reinitialisation lorsque tous les bits ont {t} faits
1340 V=V/2:IF V<1 THEN V=8:' V prend les valeurs 8 4 2 1 8 4 2 1
1350 NEXT P
1360 ' *** Calcul des 2 Words caract{risant la ligne en cours
1370 C(0,7,0)=C(0,7,0)+C(0,6,0)+C(0,5,0)+C(0,4,0)
1380 C(0,3,0)=C(0,3,0)+C(0,2,0)+C(0,1,0)+C(0,0,0)
1390 C(0,7,1)=C(0,7,1)+C(0,6,1)+C(0,5,1)+C(0,4,1)
1400 C(0,3,1)=C(0,3,1)+C(0,2,1)+C(0,1,1)+C(0,0,1)
1410 C(0,7,2)=C(0,7,2)+C(0,6,2)+C(0,5,2)+C(0,4,2)
1420 C(0,3,2)=C(0,3,2)+C(0,2,2)+C(0,1,2)+C(0,0,2)
1430 C(0,7,3)=C(0,7,3)+C(0,6,3)+C(0,5,3)+C(0,4,3)
1440 C(0,3,3)=C(0,3,3)+C(0,2,3)+C(0,1,3)+C(0,0,3)
1450 W(0+LIGNE)=C(0,7,0)+16*C(0,3,0)+256*C(0,7,1)+4096*C(0,3,1)
1460 W(8+LIGNE)=C(0,7,2)+16*C(0,3,2)+256*C(0,7,3)+4096*C(0,3,3)
1470 NEXT LIGNE
1480 PRINT:PRINT"DEFCHAR"HEX$(CA);
1490 FOR I=0 TO 15
1500 PRINT", "HEX$(W(I));
1510 NEXT I
1520 END

```

